

#T59

Транспорт сообщений обмена данными.

Программист 1С: с нуля до профи



Способы транспорта сообщений обмена

Способы транспорта сообщений обмена

Через файл:

- Локальный или сетевой каталог
- FTP-ресурс
- Почтовые сообщения (email)

Прямое подключение (COM)


- возможно когда источник и приемник находятся на одном компьютере, либо в одной локальной сети

Подключение через Интернет


- веб-сервисы

ОБМЕН ФАЙЛАМИ

Локальный или сетевой каталог



Обмен через локальный или сетевой каталог



Источник

Приемник

Программное формирование сообщений (универсальный формат)

Выгрузка данных из источника:


```
//создание экземпляра объекта обработки "УниверсальныйОбменДаннымиXML"  
Обработка = Обработки.УниверсальныйОбменДаннымиXML.Создать ();  
  
//режим обмена: Выгрузка или Загрузка  
Обработка.РежимОбмена = "Выгрузка";  
  
//период отбора объектов  
Обработка.ДатаНачала      = НачалоДня (ДобавитьМесяц (ТекущаяДата (), -1));  
Обработка.ДатаОкончания  = КонецМесяца (ТекущаяДата ());  
  
//путь к файлу правил обмена  
Обработка.ИмяфайлаПравилОбмена = "\\SERV\Exchange";  
  
//путь к файлу с данными  
Обработка.ИмяфайлаОбмена = "\\SERV\Exchange\UT_BP_Message.xml";  
  
//загрузка правил обмена  
Обработка.ЗагрузитьПравилаОбмена ();  
  
//выгрузка данных  
Обработка.ВыполнитьВыгрузку ();
```

Программное формирование сообщений (универсальный формат)

Загрузка данных в приемнике:


```
//создание экземпляра объекта обработки "УниверсальныйОбменДаннымиXML"  
Обработка = Обработки.УниверсальныйОбменДаннымиXML.Создать ();  
  
//установка режима обмена  
Обработка.РежимОбмена = "Загрузка";  
  
//путь к файлу с данными  
Обработка.ИмяфайлаОбмена = "\\SERV\Exchange\UT_BP_Message.xml";  
  
//дополнительные настройки  
Обработка.ОптимизированнаяЗаписьОбъектов = Истина;  
Обработка.ЗаписыватьРегистрыНаборамиЗаписей = Истина;  
Обработка.ЗаписыватьВИнформационнуюБазуТолькоИзмененныеОбъекты = Истина;  
Обработка.ЗагружатьДанныеВРежимеОбмена = Истина;  
  
//загрузка данных  
Обработка.ВыполнитьЗагрузку ();
```

FTP-pecypc



Что такое FTP

- **FTP** (File Transfer Protocol) – протокол передачи файлов



Параметры подключения к FTP-серверу





- Адрес
 - Пример: **vh112.hoster.by** или **93.125.99.124**
- Порт
 - Пример: **20**
- Логин
 - Пример: **1c_user**
- Пароль
 - Пример: **pa\$\$word5**

Схема обмена через FTP

Каталог временных файлов



Каталог временных файлов



**Источник
(FTP-клиент)**



**Приемник
(FTP-клиент)**

Объект для работы с FTP

FTPСоединение (FTPConnection)

Свойства:

- [ЗащищенноеСоединение \(SecureConnection\)](#)
- [Пароль \(Password\)](#)
- [ПассивныйРежим \(PassiveMode\)](#)
- [Пользователь \(User\)](#)
- [Порт \(Port\)](#)
- [Прокси \(Proxy\)](#)
- [Сервер \(Host\)](#)
- [Таймаут \(Timeout\)](#)

Методы:

- [Записать \(Put\)](#)
- [НайтиФайлы \(FindFiles\)](#)
- [Переместить \(Move\)](#)
- [Получить \(Get\)](#)
- [СоздатьКаталог \(CreateDirectory\)](#)
- [ТекущийКаталог \(GetCurrentDirectory\)](#)
- [Удалить \(Delete\)](#)
- [УстановитьТекущийКаталог \(SetCurrentDirectory\)](#)

Конструкторы:

- [По указанному серверу](#)

Описание:

Предназначен для работы с файлами и каталогами на ftp-серверах.

Для защиты соединения может быть использован протокол TLS 1.2.

Поддерживается работа по протоколу FTPS и FTPES.

Выгрузка данных из источника на FTP-сервер

```
//создание экземпляра объекта обработки "УниверсальныйОбменДаннымиXML"
Обработка = Обработки.УниверсальныйОбменДаннымиXML.Создать ();

//режим обмена: Выгрузка или Загрузка
Обработка.РежимОбмена = "Выгрузка";

//период отбора объектов
Обработка.ДатаНачала = НачалоДня(ДобавитьМесяц(ТекущаяДата(),-1));
Обработка.ДатаОкончания = КонецМесяца(ТекущаяДата());

//путь к файлу правил обмена
Обработка.ИмяфайлаПравилОбмена = "\\SERV\Exchange\ПравилаОбменаДанными.xml";

//путь к файлу с данными
Обработка.ИмяфайлаОбмена = "\\SERV\Exchange\UT_BP_Message.xml";

//загрузка правил обмена
Обработка.ЗагрузитьПравилаОбмена();

//выгрузка данных
Обработка.ВыполнитьВыгрузку();
```

```
//*****
//*****ЗАГРУЗКА ФАЙЛА НА FTP-СЕРВЕР*****
АдресСервера = "195.69.187.77";
Логин = "user23723";
Пароль = "ruvritMhqTeV";

//установка соединения с FTP-сервером
FTPСоединение = Новый FTPСоединение(АдресСервера, 21, Логин, Пароль,,Истина, 20);

Попытка
    // Установка текущего каталога на FTP-сервере
    FTPСоединение.УстановитьТекущийКаталог("/");

    // Загрузка файла на FTP-сервер в каталог exchange
    // Внимание! Используем прямые слэши.
    FTPСоединение.Записать(Обработка.ИмяфайлаОбмена, "/exchange/UT_BP_Message.xml");
Исключение
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = "Не удалось загрузить файл на FTP по причине:" + ОписаниеОшибки();
    Сообщение.Сообщить();
КонецПопытки;
```


Загрузка данных в приемник (с загрузкой файла с FTP-сервера)

```
//*****  
//*****СКАЧИВАНИЕ ФАЙЛА С FTP-СЕРВЕРА*****  
  
АдресСервера = "195.69.187.77";  
Логин = "user23723";  
Пароль = "ruvritMhQTeV";  
  
FTPСоединение = Новый FTPСоединение(АдресСервера, 21, Логин, Пароль,,Истина, 20);  
  
// Установим текущий каталог.  
FTPСоединение.УстановитьТекущийКаталог("/");  
  
ИмяВременногофайла = ПолучитьИмяВременногофайла("xml");  
  
Попытка  
    // Скачаем "exchange/UT_BP_Message.xml" в локальный каталог временных файлов  
    // Внимание! Используем прямые слэши.  
    FTPСоединение.Получить("exchange/UT_BP_Message.xml", ИмяВременногофайла);  
Исключение  
    Сообщение = Новый СообщениеПользователю;  
    Сообщение.Текст = "Не удалось скачать файл с FTP-сервера по причине: " + ОписаниеОшибки();  
    Сообщение.Поле = "";  
    Сообщение.УстановитьДанные();  
    Сообщение.Сообщить();  
    Возврат;  
КонецПопытки;  
  
//создание экземпляра объекта обработки "УниверсальныйОбменДаннымиXML"  
Обработка = Обработки.УниверсальныйОбменДаннымиXML.Создать();  
  
//установка режима обмена  
Обработка.РежимОбмена = "Загрузка";  
  
//путь к файлу с данными  
Обработка.ИмяфайлаОбмена = "\\SERV\Exchange\UT_BP_Message.xml";  
  
//дополнительные настройки  
Обработка.ОптимизированнаяЗаписьОбъектов = Истина;  
Обработка.ЗаписыватьРегистрыНаборамиЗаписей = Истина;  
Обработка.ЗаписыватьВИнформационнуюБазуТолькоИзмененныеОбъекты = Истина;  
Обработка.ЗагружатьДанныеВРежимеОбмена = Истина;  
  
//загрузка данных  
Обработка.ВыполнитьЗагрузку();
```

Почтовые сообщения (Email)


Работа с почтой. Почтовый сервер

Почтовый сервер — сервер, расположенный в Интернете или интрасети, принимающий или передающий сообщения электронной **почты** по протоколам SMTP, POP, IMAP.




Работа с почтой. Протоколы. SMTP

SMTP (Simple Mail Transfer Protocol) — используется для связи с удаленным сервером и последующей **отправке** сообщений с локального клиента на удаленный сервер, и в конечном итоге на сервер получателя сообщений




Почтовый сервер. Протоколы. POP3

POP3 (Post Office Protocol Version 3) — используется для связи с удаленным сервером электронной почты и **загрузки** сообщений на локальный почтовый клиент



Почтовый сервер. Протоколы. POP3

IMAP (Internet Message Access Protocol) — используется для связи с удаленным сервером электронной почты и **загрузки** сообщений на локальный почтовый клиент. Отличие от POP3 - загружаются только лишь заголовки электронных сообщений, сам текст письма остается на сервере



Объекты для работы с почтой в 1С

ИнтернетПочта (InternetMail)

Свойства:

[СимволРазделитель \(DelimiterChar\)](#)
[ТекущийПочтовыйЯщик \(CurrentMailbox\)](#)

Методы:

[Выбрать \(Get\)](#)
[ОбъединитьЧастичные \(CombinePartial\)](#)
[Отключиться \(Logout\)](#)
[ОтменитьПодпискуНаПочтовыйЯщик \(UnsubscribeFromMailbox\)](#)
[ОтменитьУдалениеСообщений \(UndeleteMessages\)](#)
[ОчиститьУдаленныеСообщения \(ClearDeletedMessages\)](#)
[ПереименоватьПочтовыйЯщик \(RenameMailbox\)](#)
[ПереместитьВПочтовыйЯщик \(MoveToMailbox\)](#)
[Подключиться \(Logon\)](#)
[ПодписатьсяНаПочтовыйЯщик \(SubscribeToMailbox\)](#)
[ПолучитьЗаголовки \(GetHeaders\)](#)
[ПолучитьИдентификаторы \(GetUIDL\)](#)
[ПолучитьКоличествоСообщений \(GetMessageCount\)](#)
[ПолучитьПочтовыеЯщики \(GetMailboxes\)](#)
[ПолучитьПочтовыеЯщикиПоПодписке \(GetMailboxesBySubscription\)](#)
[ПолучитьФлагиСообщений \(GetMessagesFlags\)](#)
[Послать \(Send\)](#)
[СоздатьПочтовыйЯщик \(CreateMailbox\)](#)
[УдалитьПочтовыйЯщик \(DeleteMailbox\)](#)
[УдалитьСообщения \(DeleteMessages\)](#)
[УстановитьФлагиСообщений \(SetMessagesFlags\)](#)

Конструкторы:

[По умолчанию](#)

Описание:

Предназначен для доступа к почтовым серверам для отправки и получения сообщений (писем). В отличие от объекта [Почта](#),

ИнтернетПочтовоеСообщение (InternetMailMessage)

Свойства:

[АдресаУведомленияОДоставке \(DeliveryReceiptAddresses\)](#)
[АдресаУведомленияОПрочтении \(ReadReceiptAddresses\)](#)
[Важность \(Importance\)](#)
[Вложения \(Attachments\)](#)
[ДатаОтправления \(PostingDate\)](#)
[ДатаПолучения \(DateReceived\)](#)
[Заголовок \(Header\)](#)
[Идентификатор \(UID\)](#)
[ИдентификаторСообщения \(MessageID\)](#)
[ИмяОтправителя \(SenderName\)](#)
[Категории \(Categories\)](#)
[Кодировка \(Encoding\)](#)
[Копии \(Cc\)](#)
[ОбратныйАдрес \(ReplyTo\)](#)
[Организация \(Organization\)](#)
[Отправитель \(From\)](#)
[Получатели \(To\)](#)
[Размер \(Size\)](#)
[СлепыеКопии \(Bcc\)](#)
[СмещениеДатыОтправления \(PostingDateOffset\)](#)
[СпособКодированияНеASCIIСимволов \(NonASCIISymbolsEncodingMode\)](#)
[СтатусРазбора \(ParseStatus\)](#)
[Тексты \(Texts\)](#)
[Тема \(Subject\)](#)
[УведомитьОДоставке \(RequestDeliveryReceipt\)](#)
[УведомитьОПрочтении \(RequestReadReceipt\)](#)
[Частичное \(Partial\)](#)

Методы:

[ОбработатьТексты \(ProcessTexts\)](#)
[ПолучитьИсходныеДанные \(GetSourceData\)](#)

ИнтернетПочтовыйПрофиль (InternetMailProfile)

Свойства:

[POP3ПередSMTP \(POP3BeforeSMTP\)](#)
[АдресСервераIMAP \(IMAPServerAddress\)](#)
[АдресСервераPOP3 \(POP3ServerAddress\)](#)
[АдресСервераSMTP \(SMTPServerAddress\)](#)
[АутентификацияПоТокену \(TokenAuthentication\)](#)
[ИспользоватьSSLIMAP \(IMAPUseSSL\)](#)
[ИспользоватьSSLPOP3 \(POP3UseSSL\)](#)
[ИспользоватьSSLSMTP \(SMTPUseSSL\)](#)
[Пароль \(Password\)](#)
[ПарольIMAP \(IMAPPassword\)](#)
[ПарольSMTP \(SMTPPassword\)](#)
[Пользователь \(User\)](#)
[ПользовательIMAP \(IMAPUser\)](#)
[ПользовательSMTP \(SMTPUser\)](#)
[ПортIMAP \(IMAPPort\)](#)
[ПортPOP3 \(POP3Port\)](#)
[ПортSMTP \(SMTPPort\)](#)
[Таймаут \(Timeout\)](#)
[ТокенДоступа \(AccessToken\)](#)
[ТолькоЗащищеннаяАутентификацияIMAP \(IMAPSecureAuthenticationOnly\)](#)
[ТолькоЗащищеннаяАутентификацияPOP3 \(POP3SecureAuthenticationOnly\)](#)
[ТолькоЗащищеннаяАутентификацияSMTP \(SMTPSecureAuthenticationOnly\)](#)

Конструкторы:

[По умолчанию](#)

Описание:

Набор свойств для соединения с сервером.

Как отправить почтовое сообщение из 1С

```
//описание почтового профиля (параметры подключения к почтовому серверу)
ПочтовыйПрофиль = Новый ИнтернетПочтовыйПрофиль;
ПочтовыйПрофиль.АдресСервераSMTP = "smtp.yandex.ru";
ПочтовыйПрофиль.АдресСервераIMAP = "imap.yandex.ru";
ПочтовыйПрофиль.ПортSMTP = 465;
ПочтовыйПрофиль.ПортIMAP = 993;
ПочтовыйПрофиль.АутентификацияSMTP = СпособSMTPАутентификации.Login;
ПочтовыйПрофиль.ПользовательSMTP = "its21092020";
ПочтовыйПрофиль.ПарольSMTP = "Its123456";
ПочтовыйПрофиль.Пользователь = "its21092020";
ПочтовыйПрофиль.Пароль = "Its123456";
ПочтовыйПрофиль.ВремяОжидания = 30;
ПочтовыйПрофиль.ИспользоватьSSLSMTP = Истина;

//создание почтового сообщения
Сообщение = Новый ИнтернетПочтовоеСообщение;
Сообщение.Вложения.Добавить (Обработка.ИмяФайлаОбмена);
Сообщение.Получатели.Добавить ("its21092020@yandex.ru");
Сообщение.Тема = "Обмен данными УТ-ВП";
Сообщение.ИмяОтправителя = "Обмен";
Сообщение.Отправитель = "its21092020@yandex.ru";

Текст = Сообщение.Тексты.Добавить ("файл обмена в приложении");
Текст.ТипТекста = ТипТекстаПочтовогоСообщения.ПростойТекст;

//подключение к почтовому серверу
Почта = Новый ИнтернетПочта;
Почта.Подключиться (ПочтовыйПрофиль);

//отправка почтового сообщения
Почта.Послать (Сообщение);

//отключение
Почта.Отключиться ();
```

Как получить почтовое сообщение в 1С

```
//описание почтового профиля (параметры подключения к почтовому серверу)
```

```
ПочтовыйПрофиль = Новый ИнтернетПочтовыйПрофиль;  
ПочтовыйПрофиль.АдресСервераSMTP = "smtp.yandex.ru";  
ПочтовыйПрофиль.АдресСервераIMAP = "imap.yandex.ru";  
ПочтовыйПрофиль.ПортSMTP = 465;  
ПочтовыйПрофиль.ПортIMAP = 993;  
ПочтовыйПрофиль.АутентификацияSMTP = СпособSMTPАутентификации.Login;  
ПочтовыйПрофиль.ПользовательSMTP = "its21092020";  
ПочтовыйПрофиль.ПарольSMTP = "Its123456";  
ПочтовыйПрофиль.ПользовательIMAP = "its21092020";  
ПочтовыйПрофиль.ПарольIMAP = "Its123456";  
ПочтовыйПрофиль.Пользователь = "its21092020";  
ПочтовыйПрофиль.Пароль = "Its123456";  
ПочтовыйПрофиль.ВремяОжидания = 30;  
ПочтовыйПрофиль.ИспользоватьSSLSMTP = Истина;  
ПочтовыйПрофиль.ИспользоватьSSLIMAP = Истина;
```

```
//подключение к почтовому серверу (по протоколу IMAP)
```

```
Почта = Новый ИнтернетПочта;  
Почта.Подключиться(ПочтовыйПрофиль, ПротоколИнтернетПочты.IMAP);
```

```
//получение писем
```

```
МассивПисем = Почта.Выбрать(Ложь);
```


```
//обработка полученных писем
```

```
Для Каждого Письмо из МассивПисем Цикл  
    ТемаПисьма = Письмо.Тема;  
    Для Каждого Текст из Письмо.Тексты Цикл  
        ТекстПисьма = Текст.Текст;  
    КонецЦикла;  
    Для Каждого Вложение из Письмо.Вложения Цикл  
        Вложение.Данные.Записать("C:\temp\"+Вложение.Имяфайла);  
    КонецЦикла;  
КонецЦикла;
```


Прямое подключение (COM)

Технология Automation Client (COM-подключение)

- Позволяет подключиться напрямую из одной информационной базы к другой



Технология Automation Client (COM-подключение)



Технология Automation Client (COM-подключение)

The image shows a Windows File Explorer window and a Command Prompt window. The File Explorer window is open to the path `1cv8 > 8.3.20.1613 > bin`. The file `comcntr.dll` is selected. The Command Prompt window shows the command `regsvr32 "C:\Program Files\1cv8\8.3.20.1613\bin\comcntr.dll"` being executed.

File Explorer Path: `1cv8 > 8.3.20.1613 > bin`

Имя	Дата изменения	Тип	Размер
<code>comcntr.dll</code>	09.11.2021 12:46	Расширение при...	601 KB
<code>concr140.dll</code>	15.08.2019 16:34	Расширение при...	325 KB

Command Prompt Command: `regsvr32 "C:\Program Files\1cv8\8.3.20.1613\bin\comcntr.dll"`

Технология Automation Client (COM-подключение)

Пример:

```
ПутьКБазе      = "D:\Bases_1C\AccountingT";
Пользователь   = "Администратор";
Пароль         = "";

// Подключение к файловой базе
Соединение = Новый СОМОбъект("V83.COMConnector");
ПараметрыДоступа = СтрШаблон("File=""%1"";Usr=""%2"";Pwd=""%3"";", ПутьКБазе, Пользователь, Пароль);
Попытка
    БазаПриемник = Соединение.Connect(ПараметрыДоступа);
Исключение
    Сообщить("Проблемы при подключении: " + ОписаниеОшибки());
    Возврат;
КонецПопытки;

Если БазаПриемник = Неопределено Тогда
    Возврат;
КонецЕсли;

//Пример: создание нового элемента справочника в базе-приемнике
НовыйТовар = БазаПриемник.Справочники.Номенклатура.СоздатьЭлемент();
НовыйТовар.Наименование = "Тестовый товар";
НовыйТовар.ОбменДанными.Загрузка = Истина;
НовыйТовар.Записать();
```

Конец 59-го занятия

Подписывайтесь на нас в социальных сетях 😊



<https://www.youtube.com/channel/UC124tAuP11GlaGpK3YGHzw>



<https://www.facebook.com/ironskillsby/>



[https://www.instagram.com/ironskills_ /](https://www.instagram.com/ironskills_/)